



POWER OF SIMPLICITY

Developer Manual – Cost Centre wise Profit & Loss

The information contained in this document is current as of the date of publication and subject to change. Because Tally must respond to changing market conditions, it should not be interpreted to be a commitment on the part of Tally, and Tally cannot guarantee the accuracy of any information presented after the date of publication. The information provided herein is general, not according to individual circumstances, and is not intended to substitute for informed professional advice.

This document is for informational purposes only. TALLY MAKES NO WARRANTIES, EXPRESS OR IMPLIED, IN THIS DOCUMENT AND SHALL NOT BE LIABLE FOR LOSS OR DAMAGE OF WHATEVER NATURE, ARISING OUT OF, OR IN CONNECTION WITH THE USE OF OR INABILITY TO USE THE CONTENT OF THIS PUBLICATION, AND/OR ANY CONDUCT UNDERTAKEN BY PLACING RELIANCE ON THE CONTENTS OF THIS PUBLICATION.

Complying with all applicable copyright and other intellectual property laws is the responsibility of the user. All rights including copyrights, rights of translation, etc., are vested exclusively with TALLY SOLUTIONS PRIVATE LIMITED. No part of this document may be reproduced, translated, revised, stored in, or introduced into a retrieval system, or transmitted in any form, by any means (electronic, mechanical, photocopying, recording, or otherwise), or for any purpose, without the express written permission of Tally Solutions Pvt. Ltd.

Tally may have patents, patent applications, trademarks, copyrights, or other intellectual property rights covering subject matter in this document. Except as expressly provided in any written licence agreement from Tally, the furnishing of this document does not give you any licence to these patents, trademarks, copyrights, or other intellectual property.

© 2009 Tally Solutions Pvt. Ltd. All rights reserved.

Tally.Developer 9 is either registered trademarks or trademarks of Tally Solutions Pvt. Ltd. in India and/or other countries. All other trademarks are properties of their respective owners.

Version: Developer Manual – Cost Centre wise Profit & Loss/1.0/October 2009

Cost Centre wise Profit & Loss

Scenario

A firm **Global Enterprises** using Tally.ERP 9 needs to view the profits and the loss based on their projects.

Requirement Statement

User should be able to view the expenses paid and incomes received from the project. And whether there was a profit or a loss from the project.

Solution

In order to enable cost centre wise profit and loss, the following steps have to be adhered:

I. Changes in **Profit and Loss** Report

a. Adding a button for selection of cost centre name at Form Level

```
[#FORM: Profit and Loss]

ADD      : BUTTON      : AFTER      : ChangeValue :CostCentreButton

[BUTTON: CostCentreButton]

TITLE    : "Cost Centre"
KEY      : Alt+S
ACTION   : ModifyVariables      : Cost Centre Filter Report

[REPORT: Cost Centre Filter Report]

USE      : Modify Variables
LOCAL    : PART : Modify Variables : LINES      : Cost Centre FilterLine
LOCAL    : FIELD: MV Title         : INFO     : $$LocaleString:"Cost Centre Filter"
TITLE    : $$LocaleString:"Cost Centre Filter"

[LINE: Cost Centre FilterLine]

FIELDS   : Cost Centre Filter Field

[FIELD: Cost Centre Filter Field]

USE      : Name Field
TABLE    : Cost Centre Table, All Items
SHOW TABLE : ALWAYS
MODIFIES : CostCenterFilterVar
```

b. Changes in report for variable value to set by default

```
[#REPORT: Profit and Loss]

VARIABLE: CostCenterFilterVar
SET      : CostCenterFilterVar      : $$SysName:AllItems
```

c. Changes at Part Level

```
[#PART: PLInDirExpenses]

LOCAL : FIELD : BSMainTotal      : SET AS : @@PLDNettIn
ADD    : LINES : AT BEGINNING    : PLD GL
ADD    : BOTTOM LINES : AT BEGINNING : PLD NP
LOCAL : LINE  : PLD GL: LOCAL    : FIELD : Default      : INACTIVE : @@PLDExpFilt < @@PLDIncFilt OR @@PLDExpFilt = @@PLDIncFilt
LOCAL : LINE  : PLD GL: LOCAL    : FIELD : DSP DispName : SET AS  : "Gross Loss B/f"

[LINE: PLD NP]

FIELDS : DSP AccName, PLAmt
REPEAT : PLAmt
LOCAL : FIELD : DSP DispName : SET AS : "Net Profit"
LOCAL : FIELD : BSMainAmt    : SET AS : @@PLDGPInPos
LOCAL : FIELD : Default      : INACTIVE : @@PLDInExpFiltWGL > @@PLDInIncFiltWGP OR @@PLDInExpFiltWGL = @@PLDInIncFiltWGP

[#PART: PLInDirIncome]

LOCAL : FIELD : BSMainTotal : SET AS : @@PLDNettIn
LOCAL : FIELD : BSMainTotal : COLOR : Red
ADD    : LINES : AT BEGINNING : PLD GP
ADD    : BOTTOM LINES : AT BEGINNING : PLD NL
LOCAL : LINE  : PLD GP: LOCAL    : FIELD : Default      : INACTIVE : @@PLDExpFilt > @@PLDIncFilt OR @@PLDExpFilt = @@PLDIncFilt
LOCAL : LINE  : PLD GP: LOCAL    : FIELD : DSP DispName : SET AS  : "Gross Profit B/f"

[LINE: PLD NL]

FIELDS : DSP AccName, PLAmt
REPEAT : PLAmt
LOCAL : FIELD : DSP DispName : SET AS : "Net Loss"
LOCAL : FIELD : Default      : INACTIVE : @@PLDInExpFiltWGL < @@PLDInIncFiltWGP OR @@PLDInIncFiltWGP = @@PLDInIncFilt
LOCAL : FIELD : BSMainAmt    : SET AS : @@PLDGPInPos

[#FIELD: BSMainTotal]

SET AS : @@PLDNettIn + @@PLDGPPos
```

II. The following objects have to be modified to achieve this feature.

a. Group Object

```
[#OBJECT: Group]

BSClosing : IF @@CCFilter THEN $$AsPositive:$$CCClosing:##CostCenterFilterVar ELSE $$AsPositive:$ClosingBalance
```

b. Ledger Object

```
[#OBJECT: Ledger]

BSClosing : IF @@CCFilter THEN $$AsPositive:$$CCClosing:##CostCenterFilterVar ELSE $$AsPositive:$ClosingBalance
```

c. GrossProfitCO

```
[#OBJECT: GrossProfitCO]

Name : $$LocaleString:"Gross Profit"

IsGPCOLine : YES
IsPLObject : YES
SortPosition : 3000
IsTBalObject : YES

TBalClosing : @@GPClosingProfit
BSClosing : $TBalClosing
TBalOpening : $TBalClosing
BSOpening : $TBalClosing
```

d. GrossProfitBF

```
[#OBJECT: GrossProfitBF]

Name : $$LocaleString:"Gross Profit"

IsGPBFLine : YES
IsPLObject : YES
SortPosition : 0
IsTBalObject : YES

TBalClosing : $$AsPositive:$$Negative:@@GPClosingProfit
BSClosing : $TBalClosing
TBalOpening : $TBalClosing
BSOpening : $TBalClosing
```

TDL Capabilities Used

- I. Adding a **Button** to the form
- II. Set variable values at report level
- III. Localizing fields at part level and changing values
- IV. Creating methods at object level